Pensieve header: Developing ρ_d.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Talks\\Oaxaca-2210"];
```

In[•]:=
```
Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.

Loading Rot.m from http://drorbn.net/la22/ap to compute rotation numbers.
```

In[•]:=
```
<< "../../Projects/Profile/Profile.m"
```

```
This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.

This version: April 2020. Original version: July 1994.
```

In[•]:=
```
CF[ℰ_] := Module[{vs = Union[{ϵ}, Cases[ℰ, (g | p | x)_, ∞]]}, Total[
   CoefficientRules[Expand@ℰ, vs] /. (ps_ → c_) :> Factor[c] (Times @@ vs^ps)
  ]]
```

In[•]:=
```
{p*, x*, π*, ξ*} = {π, ξ, p, x};  (u_i__)* := (u*)_i;
```

In[•]:=
```
Zip_{}[ℰ_] := ℰ;
Zip_{ξ_,ξs___}[ℰ_] := (Collect[ℰ // Zip_{ξs}, ξ] /. f_. ξ^d_. :> (D[f, {ξ*, d}])) /. ξ* → 0
```

In[•]:=
```
V@γ_{d_,0}[j_] := 0; V@γ_{1,φ_}[k_] := φ (1/2 - p_k x_k);
```

In[•]:=
```
V@r_{1,s_}[i_, j_] :=
  Expand[s (-1/2 + p_i x_i - p_j x_i + 1/2 (-1 + T^s) p_i p_j x_i^2 + 1/2 (1 - T^s) p_j^2 x_i^2 - p_i p_j x_i x_j + p_j^2 x_i x_j)];
```

In[•]:=
```
V@γ_{2,1}[k_] := -1/2 p_k x_k; V@γ_{2,-1}[k_] := -1/2 p_k x_k;
```

In[•]:=
```
V@r_{2,1}[i_, j_] := -1/2 p_i x_i + (p_j x_i)/2 + 1/4 (1 - 3 T) p_i p_j x_i^2 + 1/4 (-1 + 3 T) p_j^2 x_i^2 + 1/3 (-1 + T) p_i^2 p_j x_i^3 -
   1/6 (-1 + T) (5 + T) p_i p_j^2 x_i^3 + 1/6 (-1 + T) (3 + T) p_j^3 x_i^3 + 3/2 p_i p_j x_i x_j - 3/2 p_j^2 x_i x_j -
   1/2 p_i^2 p_j x_i^2 x_j + 1/2 (2 + T) p_i p_j^2 x_i^2 x_j + 1/2 (-1 - T) p_j^3 x_i^2 x_j - 1/2 p_i p_j^2 x_i x_j^2 + 1/2 p_j^3 x_i x_j^2;
```

In[•]:=
```
V@r_{2,-1}[i_, j_] := -1/2 p_i x_i + p_j x_i/2 + (-3+T) p_i p_j x_i^2/(4 T) - (-3+T) p_j^2 x_i^2/(4 T) - (-1+T) p_i^2 p_j x_i^3/(3 T) +
  (-1+T) (1+5 T) p_i p_j^2 x_i^3/(6 T^2) - (-1+T) (1+3 T) p_j^3 x_i^3/(6 T^2) + 3/2 p_i p_j x_i x_j - 3/2 p_j^2 x_i x_j -
  1/2 p_i^2 p_j x_i^2 x_j + (1+2 T) p_i p_j^2 x_i^2 x_j/(2 T) - (1+T) p_j^3 x_i^2 x_j/(2 T) - 1/2 p_i p_j^2 x_i x_j^2 + 1/2 p_j^3 x_i x_j^2;
```

In[•]:=
```
gPair[1] = 1;
gPair[ε^(d_)· Bs_] := ε^d gPair[Bs];
gPair[c_?NumberQ * Bs_] := c gPair[Bs];
gPair[ℰ_Plus] := gPair /@ ℰ;
gPair[r_{d_,s_}[i_, j_]^(p_)·] := gPair[{r_{d,s}[i, j]^p}];
gPair[γ_{d_,φ_}[k_]^(p_)·] := gPair[{γ_{d,φ}[k]^p}];
gPair[Bs_Times] := gPair[List @@ Bs];
gPair[Bs_List] := Module[{es, BBs, res},
  BBs = Bs /. e_^(p_) :> Sequence @@ Table[e, {p}];
  es = Union @@ (List @@@ BBs);
  res = Simplify@Zip_{Join@@Table[{p_{1,α},p_{2,α},x_{1,α},x_{2,α}},{α,es}}[Times[
      Times @@ (BBs /. {
          r_{d_,s_}[i_, j_] :> (V[r_{d,s}[i, j]] /. {p_i → p_{2,i}, p_j → p_{2,j}, x_i → x_{2,i}, x_j → x_{2,j}}),
          γ_{d_,φ_}[k_] :> (V[γ_{d,φ}[k]] /. {p_k → p_{1,k}, x_k → x_{1,k}})
        }),
      Exp[Sum[g_{α,β} (π_{1,α} + π_{2,α}) (ξ_{1,β} + ξ_{2,β}), {α, es}, {β, es}] - Sum[ξ_{1,α} π_{2,α}, {α, es}]]
    ]]
 ]
```

In[•]:=
```
r_{1,-1}[1, 2] // FullForm
```

Out[•]//FullForm=
```
Subscript[r, 1, -1][1, 2]
```

In[•]:=
```
gPair[r_{1,-1}[3, 4]]
```

» {{r_{1,-1}[3, 4]}, {r_{1,-1}[3, 4]}, {3, 4}}

Out[•]=
$$\frac{1}{2}+\left(-1+\frac{1}{T}\right) g_{4,3}^2+g_{4,3}\,(1+g_{3,4}-2\,g_{4,4})+g_{3,3}\left(-1+\frac{(-1+T)\,g_{4,3}}{T}+g_{4,4}\right)$$

In[•]:=
```
gPair[γ_{1,-1}[3]]
```

Out[•]=
$$-\frac{1}{2}+g_{3,3}$$

In[•]:=
```
gPair[γ1,0[6]]
```
```
» {{γ1,0[6]}, {γ1,0[6]}, {6}}
```
Out[•]=
$$\{\{0\},\{0\},\{0\}\}$$

In[•]:=
```
gPair[γ1,-1[3]^2]
```
```
» {γ1,-1[3]^2}
```
Out[•]=
$$\frac{1}{4}-g_{3,3}+2\,g_{3,3}^2$$

In[•]:=
```
gPair[r2,1[3, 4] γ1,-1[3]]
```
Out[•]=
$$64\,g_{3,3}^3\,(16\,(-1+T)\,g_{4,3}-3\,g_{4,4})-\frac{1}{2}\,g_{4,3}\,\Big(1+64\,g_{3,4}^2\,g_{4,3}+8\,\left(-3+2\,T+T^2\right)\,g_{4,3}^2+2\,g_{3,4}\,\left(15+48\,(1+T)\,g_{4,3}^2+4\,g_{4,3}\,(16+5\,T-12\,g_{4,4})-20\,g_{4,4}\right)-6\,g_{4,4}+6\,g_{4,4}^2-2\,g_{4,3}\,(1-3\,T+6\,(1+T)\,g_{4,4})\Big)-8\,g_{3,3}^2\,\left(2+32\,\left(-5+4\,T+T^2\right)\,g_{4,3}^2-15\,g_{4,4}+4\,g_{4,4}^2-6\,g_{4,3}\,(7-9\,T-12\,g_{3,4}+4\,(2+T)\,g_{4,4})\right)+g_{3,3}\,\left(128\,\left(-3+2\,T+T^2\right)\,g_{4,3}^3+8\,g_{4,3}^2\,\left(-28+29\,T+5\,T^2+24\,(2+T)\,g_{3,4}-18\,(1+T)\,g_{4,4}\right)+5\,\left(1-3\,g_{4,4}+2\,g_{4,4}^2\right)-2\,g_{4,3}\,\left(1-15\,T+4\,(16+5\,T)\,g_{4,4}-24\,g_{4,4}^2+8\,g_{3,4}\,(-15+8\,g_{4,4})\right)\right)$$

In[•]:=
```
ρd_[K_] := PPρd@Module[{Cs, φ, n, A, s, i, j, k, Δ, G, d1, ρd},
  PP"Green"[
   {Cs, φ} = Rot[K]; n = Length[Cs];
   A = IdentityMatrix[2 n + 1];
   Cases[Cs, {s_, i_, j_} :> (A[[{i, j}, {i + 1, j + 1}]] += ({{-T^s, T^s - 1}, {0, -1}}))];
   Δ = T^((-Total[φ] - Total[Cs[[All, 1]]])/2) Det[A];
   G = Inverse[A];
  ];
  ρd = PPPairing@gPair[Series[Exp[
        Total[Cases[Cs, {s_, i_, j_} :> Sum[ϵ^d1 r_{d1,s}[i, j], {d1, d}]]]
         + Sum[ϵ^d1 γ_{d1,φ[[k]]}[k], {k, 2 n}, {d1, d}]
       ], {ϵ, 0, d}] // Normal // Expand];
  PPRenormalizing[
   ρd = CoefficientList[Δ Normal[Series[ρd, {ϵ, 0, d}]] /. ϵ → Δ ϵ, ϵ, d + 1]];
  PPSubstitution@Factor[ρd /. α_^+ :> α + 1 /. g_{α_,β_} :> G[[α, β]]]
 ];
```

In[•]:=
```
ρ1[Knot[3, 1]]
```

Out[•]=

$$\left\{\frac{1-T+T^2}{T},\ \frac{(-1+T)^2\left(1+T^2\right)}{T^2}\right\}$$

In[•]:=
```
Table[K → ρ1[K], {K, AllKnots[{3, 6}]}]
```

Out[•]=

$$\left\{\text{Knot}[3,1]\to\left\{\frac{1-T+T^2}{T},\ \frac{(-1+T)^2\left(1+T^2\right)}{T^2}\right\},\ \text{Knot}[4,1]\to\left\{-\frac{1-3T+T^2}{T},\ 0\right\},\right.$$

$$\text{Knot}[5,1]\to\left\{\frac{1-T+T^2-T^3+T^4}{T^2},\ \frac{(-1+T)^2\left(1+T^2\right)\left(2+T^2+2T^4\right)}{T^4}\right\},$$

$$\text{Knot}[5,2]\to\left\{\frac{2-3T+2T^2}{T},\ \frac{(-1+T)^2\left(5-4T+5T^2\right)}{T^2}\right\},$$

$$\text{Knot}[6,1]\to\left\{-\frac{(-2+T)(-1+2T)}{T},\ \frac{(-1+T)^2\left(1-4T+T^2\right)}{T^2}\right\},$$

$$\text{Knot}[6,2]\to\left\{-\frac{1-3T+3T^2-3T^3+T^4}{T^2},\ \frac{(-1+T)^2\left(1-4T+4T^2-4T^3+4T^4-4T^5+T^6\right)}{T^4}\right\},$$

$$\left.\text{Knot}[6,3]\to\left\{\frac{1-3T+5T^2-3T^3+T^4}{T^2},\ 0\right\}\right\}$$

In[•]:=
```
ρ2[Knot[3, 1]]
```

Out[•]=

$$\left\{\frac{1-T+T^2}{T},\ \frac{(-1+T)^2\left(1+T^2\right)}{T^2},\ \frac{1-4T+7T^2-12T^3+18T^4-12T^5+7T^6-4T^7+T^8}{2T^3\left(1-T+T^2\right)}\right\}$$

In[•]:=
```
BeginProfile[]
Timing[z1 = ρ2[Knot[10, 106]]]
PrintProfile[]
```

Out[•]=

ProfileRoot

Out[•]=

$$\Big\{8.46875,\ \Big\{-\frac{\left(1-T+T^2\right)\left(-1+T-2T^2+T^3\right)\left(-1+2T-T^2+T^3\right)}{T^4},\ -\frac{1}{T^8}(-1+T)^2\left(1-6T+20T^2-48T^3+82T^4-114T^5+134T^6-140T^7+134T^8-114T^9+82T^{10}-48T^{11}+20T^{12}-6T^{13}+T^{14}\right),\ -\Big(\left(1-16T+127T^2-676T^3+2735T^4-8980T^5+24\,938T^6-60\,420T^7+131\,072T^8-259\,992T^9+477\,614T^{10}-814\,576T^{11}+1\,282\,448T^{12}-1\,846\,716T^{13}+2\,411\,126T^{14}-2\,836\,312T^{15}+2\,995\,252T^{16}-2\,836\,312T^{17}+2\,411\,126T^{18}-1\,846\,716T^{19}+1\,282\,448T^{20}-814\,576T^{21}+477\,614T^{22}-259\,992T^{23}+131\,072T^{24}-60\,420T^{25}+24\,938T^{26}-8980T^{27}+2735T^{28}-676T^{29}+127T^{30}-16T^{31}+T^{32}\right)\Big/\left(2T^{12}\left(1-T+T^2\right)\left(-1+T-2T^2+T^3\right)\left(-1+2T-T^2+T^3\right)\right)\Big)\Big\}\Big\}$$

Out[•]=
```
ProfileRoot is root. Profiled time: 8.469
  ( 1)     0/ 8.469 above ρd
Renormalizing: called 1 times, time in 6.203/6.203
  ( 1)  6.203/ 6.203 under ρd
Substitution: called 1 times, time in 1.485/1.485
  ( 1)  1.485/ 1.485 under ρd
Green: called 1 times, time in 0.672/0.672
  ( 1)  0.672/ 0.672 under ρd
Pairing: called 1 times, time in 0.109/0.109
  ( 1)  0.109/ 0.109 under ρd
ρd: called 1 times, time in 0./8.469
  ( 1)     0/ 8.469 under ProfileRoot
  ( 1)  0.672/ 0.672 above Green
  ( 1)  0.109/ 0.109 above Pairing
  ( 1)  6.203/ 6.203 above Renormalizing
  ( 1)  1.485/ 1.485 above Substitution
```

In[•]:=
```
BeginProfile[]
Timing[z2 = ρ2[Knot[12, NonAlternating, 369]]]
PrintProfile[]
```

Out[•]=
```
ProfileRoot
```

KnotTheory: Loading precomputed data in KnotTheory/12N.dts.

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

Out[•]=
```
$Aborted
```

Out[•]=
```
ProfileRoot is root. Profiled time: 1.407
  ( 1)    0/   0 above ρd
Green: called 1 times, time in 1.407/1.407
  ( 1)  1.410/ 1.410 under ρd
ρd: called 1 times, time in 0./0.
  ( 1)    0/   0 under ProfileRoot
  ( 1)  1.410/ 1.410 above Green
  ( 1)    0/   0 above Pairing
Pairing: called 1 times, time in 0./0.
  ( 1)    0/   0 under ρd
```

In[•]:=
```
Simplify[Thread[z1 == z2]]
```